Shader "Unlit/Shadow"
{
    Properties
    {
        _MainTex ("Texture", 2D) = "white" {}
    }
    SubShader
    {
        Pass
        {
            Tags
            {
                "RenderType"="Opaque"
            }
            LOD 100
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #pragma multi_compile_fog

            #include "UnityCG.cginc"

            struct appdata
            {
                float4 vertex : POSITION;
                float2 uv : TEXCOORD0;
            };

            struct v2f
            {
                float2 uv : TEXCOORD0;
                UNITY_FOG_COORDS(1)
                float4 vertex : SV_POSITION;
            };

            sampler2D _MainTex;
            float4 _MainTex_ST;

            v2f vert(appdata v)
            {
                v2f o;
                o.vertex = UnityObjectToClipPos(v.vertex);
                o.uv = TRANSFORM_TEX(v.uv, _MainTex);
                UNITY_TRANSFER_FOG(o, o.vertex);
                return o;
            }

            fixed4 frag(v2f i) : SV_Target
            {
                fixed4 col = tex2D(_MainTex, i.uv);
                UNITY_APPLY_FOG(i.fogCoord, col);
                return col;
            }
            ENDCG
        }

        //阴影映射
        Pass
        {
            Tags
            {
                "LightMode"="ShadowCaster"
            }
            LOD 100
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #pragma multi_compile_fog

            //告诉编译器生成多个shader变体
            //用于支持不同类型的阴影(sm sssm)
            //可以确保着色器能够在所有可能的阴影投射模式下正确渲染
            #pragma multi_compile_shadowcaster


            //包含阴影计算的宏
            #include "UnityCG.cginc"


            struct v2f
            {
                // 顶点到片元着色器阴影投射结构体数据宏
                // 这个宏定义了一些标准的成员变量
                // 这些变量用于在阴影投射路径中传递顶点数据到片元着色器中
                V2F_SHADOW_CASTER;
            };

            sampler2D _MainTex;
            float4 _MainTex_ST;

            v2f vert(appdata_base v)
            {
                v2f o;
                //转移阴影投射器法线偏移
                //用于在顶点着色器中计算和传递投射所需的变量
                //主要用途
                //1.将对象空间的顶点位置转换为裁剪空间的位置
                //2.考虑法线偏移，以减轻阴影失真问题，尤其是在处理自阴影时
                //3.传递顶带你的投影空间位置，用于后续的阴影计算
                //主要在顶点着色器中使用
                TRANSFER_SHADOW_CASTER_NORMALOFFSET(o);
                return o;
            }

            fixed4 frag(v2f i) : SV_Target
            {
                //阴影投射片元宏
                //将深度值写入到阴影映射纹理中
                //主要在片元着色器中使用
                SHADOW_CASTER_FRAGMENT(i);
            }
            ENDCG
        }
    }
}